
OIL & GAS FIELD OUTLINES FROM BUFFERED WELLS 

The VBA Code below builds oil & gas field boundary outlines (polygons) from buffered wells (points). Input well points layer must be a feature class (FC) with the following attributes:

1) Field_name

2) Buffer distance (can be unique for each well to represent reservoirs with different drainage radii) …see figure below.

Copy the code into a new module. 

Inputs:

1) In ArcMap, data frame named  “Task 1”

2) Well FC as first layer (layer 0).

Output:

Polygon feature class in same GDB as the well points FC, with one polygon field record (may be multiple polygon rings) per field_name. Overlapping buffers for the same field name are dissolved and unioned (see figure below). Adds an attribute PCTFEDLAND which can be populated using the VBA code in PCT_FED_ LAND.doc


V


Examples of buffers by reservoir (below) and unioned together as one field (next page).

Code by Kirk Kuykendall, AmberGIS; text by Sam Limerick, Z, Inc.

===================================================================

Option Explicit

Const cFEETPERMETER As Double = 3.281

'

' deviation factor is applied to the buffer distance

' so with a buffer distance of 100 and factor of 0.05 the

' resulting linear piecewise buffer would deviate from the

' original circular buffer by no more than 5 units

' The buffer distance could be increased slightly

' so that the linear piecewise buffer area is closer to the

' circle, currently this area is always smaller.

'

Const cMAXDEVIATIONFACTOR As Double = 0.01

Sub Task1()

    ' inputs: Well point featureclass in Layer (0)

    ' outputs: Polygon featureclass with one polygon for each

    '          oil & gas field.  Where each well in the field has been

    '          buffered by its buffer distance attribute and

    '          unioned together.

    '

    Dim pMxDoc As IMxDocument

    Set pMxDoc = ThisDocument

    Dim pMap As IMap

    Set pMap = GetMap("Task 1")

    Dim pWellFLayer As IFeatureLayer

    Set pWellFLayer = pMap.Layer(0)

    ' get the column name that has the oil & gas field names

    Dim sOilFldColName As String

    sOilFldColName = GetColumn(pWellFLayer.FeatureClass, _

                     "Choose the Column that holds Oil Field Names in " & pMap.Layer(0).Name, _

                     "FLDNM_HST")

    If sOilFldColName = "" Then Exit Sub

    Debug.Print sOilFldColName

    ' get the column that has the buffer distances

    Dim sBuffDistColName As String

    sBuffDistColName = GetColumn(pWellFLayer.FeatureClass, _

                       "Choose the column that has the buffer distance", _

                       "BUFFER")

    ' get the output that we'll use (set by reference)

    Dim pOutFWS As IFeatureWorkspace, sOutFCName As String

    GetOutPut pOutFWS, sOutFCName, "Save output as ..."

    If pOutFWS Is Nothing Then Exit Sub

    Dim pSR As ISpatialReference

    Set pSR = GetSR(pWellFLayer.FeatureClass)

    Dim pWellFC As IFeatureClass

    Set pWellFC = pWellFLayer.FeatureClass

    Dim pOilFldNameFld As IField

    Set pOilFldNameFld = pWellFC.Fields.Field(pWellFC.FindField(sOilFldColName))

    Dim pColumns As IFields

    Set pColumns = MakeColumns(pOilFldNameFld, pSR)

    If pColumns Is Nothing Then Exit Sub

    Dim pUID As New UID

    pUID.Value = "esriCore.Feature"

    Dim pOutFC As IFeatureClass

    Set pOutFC = pOutFWS.CreateFeatureClass(sOutFCName, pColumns, _

                                            pUID, Nothing, esriFTSimple, _

                                            "Shape", "")

    Dim pFSO As New Scripting.FileSystemObject

    Dim pTxtStream As Scripting.TextStream

    Set pTxtStream = pFSO.OpenTextFile(GetWSDir(pOutFWS) & "\" & sOutFCName & "BufferErrorLog.txt", ForWriting, True)

    BufferFeatClass pWellFLayer, sOilFldColName, sBuffDistColName, pOutFC, pTxtStream

    pTxtStream.Close

    If pOutFC Is Nothing Then

        MsgBox "unable to create output : " & sOutFCName

    End If

    Application.StatusBar.Message(0) = "Task 1 finished"

End Sub

Function GetWSDir(pWorkspace As IWorkspace) As String

    Dim l As Long

    l = InStrRev(pWorkspace.PathName, "\")

    GetWSDir = Mid(pWorkspace.PathName, 1, l - 1)

End Function

Sub BufferFeatClass(pWellFLayer As IFeatureLayer, sOilFldColName As String, _

                     sBuffDistColName As String, pOutFC As IFeatureClass, _

                     Optional pTxtStream As Scripting.TextStream = Nothing)

    Dim pDispTable As IDisplayTable

    Set pDispTable = pWellFLayer

    Dim pQF As IQueryFilter

    Application.StatusBar.Message(0) = "sorting " & pWellFLayer.FeatureClass.AliasName & "..."

    Dim pInFCur As IFeatureCursor

    Set pInFCur = GetSortedCur(pWellFLayer.FeatureClass, _

                               pDispTable.SearchDisplayTable(pQF, False), _

                               sOilFldColName)

    Dim pWellFC As IFeatureClass

    Set pWellFC = pWellFLayer.FeatureClass

    Dim lOilFld As Long

    lOilFld = pWellFC.FindField(sOilFldColName)

    Dim lBuffFld As Long

    lBuffFld = pWellFC.FindField(sBuffDistColName)

    Dim sOilField As String

    sOilField = ""

    Dim pOutGeom As ITopologicalOperator2

    Dim pOutFCur As IFeatureCursor

    Set pOutFCur = pOutFC.Insert(True)

    Dim pFBuff As IFeatureBuffer

    Set pFBuff = pOutFC.CreateFeatureBuffer

    Dim lWellCount As Long, lWellCountFld As Long

    lWellCountFld = pOutFCur.FindField("WellCount")

    If lWellCountFld = -1 Then

        MsgBox "Wellcount column not found on " & pOutFC.AliasName

        Exit Sub

    End If

    lWellCount = 0

    Dim lAcreageFld As Long

    lAcreageFld = pOutFCur.FindField("Acreage")

    If lAcreageFld = -1 Then

        MsgBox "Acreage column not found on " & pOutFC.AliasName

        Exit Sub

    End If

    Dim pInFeat As IFeature, lBadUnions As Long, lTry As Long, lTries As Long

    lBadUnions = 0

    lTries = 3 ' number of times to re-try unioning

    Dim pDict As New Scripting.Dictionary

    Dim lFeat As Long, lFeats As Long

    lFeat = 0

    'lFeats = pDispTable.DisplayTable.RowCount(pQF) bug

    lFeats = GetRowCount(pDispTable, pQF)

    For lFeat = 1 To lFeats

        Set pInFeat = pInFCur.NextFeature

        If pInFeat.Value(lOilFld) <> sOilField Or lFeat = lFeats Then

            If Not pOutGeom Is Nothing Then

                ' insert the feature

                If pDict.Count > 0 Then

                    For lTry = 1 To lTries

                        TryAgain pOutGeom, pDict, lTry, sOilField

                        If pDict.Count = 0 Then Exit For

                    Next lTry

                End If

                If pDict.Count > 0 Then

                    lBadUnions = lBadUnions + pDict.Count

                    WriteErrors pDict, pTxtStream, sOilField, lTries

                    pDict.RemoveAll

                End If

                InsertFeature pOutFCur, pOutGeom, pFBuff, _

                              lWellCountFld, lWellCount, lAcreageFld, pDict

                Application.StatusBar.Message(0) = sOilField

            Else

                ' nothing to insert yet

            End If

            sOilField = pInFeat.Value(lOilFld)

            Set pOutGeom = BufferShape(pInFeat.ShapeCopy, pInFeat.Value(lBuffFld) / cFEETPERMETER)

            CopyColumns pFBuff, pInFeat

            lWellCount = 1

        Else

            Dim pGeom As IGeometry

            Set pGeom = BufferShape(pInFeat.ShapeCopy, pInFeat.Value(lBuffFld) / cFEETPERMETER)

            Dim pUnionGeom As IGeometry

            On Error Resume Next

            Set pUnionGeom = pOutGeom.Union(pGeom)

            If Err.Number <> 0 Then

                pDict.Add pInFeat.OID, pGeom

            Else

                Set pOutGeom = pUnionGeom

            End If

            On Error GoTo 0

            lWellCount = lWellCount + 1

        End If

        Application.StatusBar.Message(0) = lFeat & " of " & lFeats & ", " & _

                                           sOilField & "," & lWellCount & _

                                           ", errorcount: " & pDict.Count

    Next lFeat

    If lBadUnions > 0 Then

        MsgBox lBadUnions & " written to error log "

    End If

End Sub

Function GetRowCount(pDispTable As IDisplayTable, pQF As IQueryFilter) As Long

    '

    ' Apparently there is a bug in pDispTable.DisplayTable.RowCount(pqf).

    ' so this is a workaround.

    '

    Dim pFCur As IFeatureCursor

    Set pFCur = pDispTable.SearchDisplayTable(pQF, False)

    Dim l As Long

    Dim pFeat As IFeature

    Set pFeat = pFCur.NextFeature

    Do Until pFeat Is Nothing

        l = l + 1

        Set pFeat = pFCur.NextFeature

    Loop

    GetRowCount = l

End Function

Sub WriteErrors(pDict As Scripting.Dictionary, pTxtStream As Scripting.TextStream, _

                sOilField As String, lTries As Long)

    Debug.Print pDict.Count & " still bad after " & lTries & " attempts for " & sOilField

    Dim vOID As Variant

    For Each vOID In pDict.Keys

        If Not pTxtStream Is Nothing Then

            pTxtStream.WriteLine sOilField & CStr(vOID)

        End If

    Next vOID

End Sub

Sub TryAgain(pOutGeom As IGeometry, pDict As Scripting.Dictionary, _

             lTry As Long, sOilField As String)

    Dim vOID As Variant, lFixed As Long

    lFixed = 0

    Dim lTotal As Long, lStillBroken As Long

    lStillBroken = 0

    lTotal = pDict.Count

    For Each vOID In pDict.Keys

        Simplify pOutGeom

        Dim pInGeom As IGeometry

        Set pInGeom = pDict.Item(vOID)

        Simplify pInGeom

        Dim pTopoOp As ITopologicalOperator

        Set pTopoOp = pOutGeom

        Dim pUnionGeom As IGeometry

        On Error Resume Next

        Set pUnionGeom = pTopoOp.Union(pInGeom)

        If Err.Number = 0 Then

            lFixed = lFixed + 1

            Set pOutGeom = pUnionGeom

            pDict.Remove vOID

        Else

            lStillBroken = lStillBroken + 1

        End If

        On Error GoTo 0

        Application.StatusBar.Message(0) = lFixed & " of " & lTotal & _

                                          " fixed for " & sOilField & _

                                          ", " & lStillBroken & " still broken"

    Next vOID

End Sub

Sub InsertFeature(pOutFCur As IFeatureCursor, pOutGeom As IGeometry, _

                  pFBuff As IFeatureBuffer, lWellCountFld As Long, _

                  lWellCount As Long, lAcreageFld As Long, _

                  pDict As Scripting.Dictionary)

    Set pFBuff.Shape = pOutGeom

    pFBuff.Value(lWellCountFld) = lWellCount

    pFBuff.Value(lAcreageFld) = GetAcreage(pOutGeom)

    pFBuff.Value(pFBuff.Fields.FindField("ErrorCount")) = pDict.Count

    pOutFCur.InsertFeature pFBuff

End Sub

Sub CopyColumns(pFBuff As IFeatureBuffer, pFeat As IFeature)

    '

    ' copy all editable non-geometry fields from pFeat to

    ' pFBuff, based on matching column name

    '

    Dim l As Long

    For l = 0 To pFBuff.Fields.FieldCount - 1

        If pFBuff.Fields.Field(l).Editable And _

           pFBuff.Fields.Field(l).Type <> esriFieldTypeGeometry Then

                Dim lInFld As Long

                lInFld = pFeat.Fields.FindField(pFBuff.Fields.Field(l).Name)

                If lInFld <> -1 Then

                    pFBuff.Value(l) = pFeat.Value(lInFld)

                End If

        End If

    Next l

End Sub

Function GetAcreage(pArea As IArea) As Double

    Dim lSqFeetPerAcre As Long

    lSqFeetPerAcre = 43560

    Dim dFeetPerMeter As Double

    dFeetPerMeter = 3.2808

    GetAcreage = (pArea.Area * 3.2808 * 3.2808) / lSqFeetPerAcre

End Function

Function BufferShape(pTopoOp As ITopologicalOperator, dDist As Double) As IPolygon

    Dim pPolyCurve As IPolycurve

    Set pPolyCurve = pTopoOp.Buffer(dDist)

    pPolyCurve.Densify pPolyCurve.Length, dDist * cMAXDEVIATIONFACTOR

    Set BufferShape = pPolyCurve

    Simplify BufferShape

End Function

Private Sub Simplify(pTopoOp2 As ITopologicalOperator2)

    pTopoOp2.IsKnownSimple = False

    pTopoOp2.Simplify

End Sub

Function GetSortedCur(pFC As IFeatureClass, pFCur As IFeatureCursor, _

                     sFldName As String) As IFeatureCursor

    Dim pTS As ITableSort

    Set pTS = New TableSort

    Set pTS.Table = pFC

    Set pTS.Cursor = pFCur

    pTS.Fields = sFldName

    pTS.Ascending(sFldName) = True

    pTS.Sort Nothing

    Set GetSortedCur = pTS.Rows

End Function

Function GetSR(pFC As IFeatureClass) As ISpatialReference

    ' get the spatial reference from the geometry column

    Set GetSR = pFC.Fields.Field(pFC.FindField(pFC.ShapeFieldName)).GeometryDef.SpatialReference

End Function

Function MakeColumns(pOilFldName As IField, pSR As ISpatialReference) As IFields

    ' define the columns that will be on the output featureclass

    Dim pFldsEdit As IFieldsEdit

    Set pFldsEdit = New Fields

    pFldsEdit.AddField pOilFldName

    pFldsEdit.AddField MakeColumn("FLDCD_DOE", esriFieldTypeDouble)

    Dim pGeomDef As IGeometryDef

    Set pGeomDef = MakeGeomDef(esriGeometryPolygon, pSR)

    pFldsEdit.AddField MakeColumn("ObjectID", esriFieldTypeOID)

    pFldsEdit.AddField MakeColumn("Shape", esriFieldTypeGeometry, 0, pGeomDef)

    pFldsEdit.AddField MakeColumn("Acreage", esriFieldTypeDouble)

    pFldsEdit.AddField MakeColumn("WellCount", esriFieldTypeInteger)

    pFldsEdit.AddField MakeColumn("ErrorCount", esriFieldTypeInteger)

    pFldsEdit.AddField MakeColumn("PctFedLand", esriFieldTypeDouble)

    Set MakeColumns = pFldsEdit

End Function

Function MakeGeomDef(lType As esriGeometryType, pSR As ISpatialReference) As IGeometryDef

    Dim pGeomDefEdit As IGeometryDefEdit

    Set pGeomDefEdit = New GeometryDef

    Set pGeomDefEdit.SpatialReference = pSR

    pGeomDefEdit.GeometryType = lType

    Set MakeGeomDef = pGeomDefEdit

End Function

Function MakeColumn(sName As String, fType As esriFieldType, _

                    Optional lLen As Long = 0, Optional pGeomDef As IGeometryDef)

    Dim pFldEdit As IFieldEdit

    Set pFldEdit = New Field

    pFldEdit.Name = sName

    pFldEdit.Type = fType

    If lLen <> 0 Then

        pFldEdit.Length = lLen

    End If

    If fType = esriFieldTypeGeometry Then

        Set pFldEdit.GeometryDef = pGeomDef

    End If

    Set MakeColumn = pFldEdit

End Function

Sub GetOutPut(pWS As IWorkspace, sName As String, sTitle As String)

    Set pWS = Nothing

    sName = ""

    Dim pGxDialog As IGxDialog

    Set pGxDialog = New GxDialog

    Set pGxDialog.ObjectFilter = New GxFilterPolygonFeatureClasses

    pGxDialog.Title = sTitle

    If pGxDialog.DoModalSave(0) Then

        Dim pGxDatabase As IGxDatabase

        If TypeOf pGxDialog.FinalLocation Is IGxDataset Then

            Set pGxDatabase = pGxDialog.FinalLocation.Parent

        Else

            Set pGxDatabase = pGxDialog.FinalLocation

        End If

        Dim pDS As IDataset

        Set pDS = OpenFC(pGxDatabase.Workspace, pGxDialog.Name)

        If Not pDS Is Nothing Then

            If MsgBox("Overwrite existing featureclass " _

                       & pGxDialog.Name & "?", vbYesNoCancel) <> vbYes Then

                Exit Sub

            Else

                If pDS.CanDelete Then

                    pDS.Delete

                Else

                    MsgBox "unable to delete " & pDS.Name

                End If

            End If

            Set pDS = Nothing

        End If

        Set pWS = pGxDatabase.Workspace

        sName = pGxDialog.Name

    End If

End Sub

Function OpenFC(pFWS As IFeatureWorkspace, sName As String) As FeatureClass

    On Error GoTo EH

    Dim pFC As IFeatureClass

    Set pFC = pFWS.OpenFeatureClass(sName)

    Set OpenFC = pFC

    Exit Function

EH:

    Set OpenFC = Nothing

End Function

Function GetColumn(pFC As IFeatureClass, sTitle As String, _

                Optional sDefault As String = "") As String

    Dim pLstDlg As IListDialog

    Set pLstDlg = New ListDialog

    Dim l As Long, lDefault As Long

    For l = 0 To pFC.Fields.FieldCount - 1

        pLstDlg.AddString pFC.Fields.Field(l).Name

        If sDefault <> "" Then

            If UCase(pFC.Fields.Field(l).Name) = UCase(sDefault) Then

                lDefault = l

            End If

        End If

    Next l

    If pLstDlg.DoModal(sTitle, lDefault, 0) Then

        GetColumn = pFC.Fields.Field(pLstDlg.Choice).Name

    Else

        GetColumn = ""

    End If

End Function

Function GetGxObj(pGxFilter As IGxObjectFilter, sTitle As String) As IUnknown

    Dim pGxDialog As IGxDialog

    Set pGxDialog = New GxDialog

    pGxDialog.Title = sTitle

    Set pGxDialog.ObjectFilter = pGxFilter

    pGxDialog.AllowMultiSelect = False

    Dim pEnumGxObj As IEnumGxObject

    If pGxDialog.DoModalOpen(0, pEnumGxObj) Then

        Dim pGxObj As IGxObject

        Set pGxObj = pEnumGxObj.Next

        If TypeOf pGxObj Is IGxDatabase Then

            Dim pGxDatabase As IGxDatabase

            Set pGxDatabase = pGxObj

            Set GetGxObj = pGxDatabase.Workspace

        ElseIf TypeOf pGxObj Is IGxDataset Then

            Dim pGxDataset As IGxDataset

            Set pGxDataset = pGxObj

            Set GetGxObj = pGxDataset.Dataset

        Else

            MsgBox "GetGxObj can't handle this object: " & pGxObj.Name

        End If

    Else

        Set GetGxObj = Nothing

    End If

End Function

The next 2 slides show the buffered field boundary for a field with two reservoirs. 



This slide displays buffers by reservoir:

		reservoir A is composed of oil wells with 80 acre buffers

		reservoir B contains oil wells (160 acre buffers) and gas wells (640 acre buffers). 

The final product of the field boundary creation process with buffers for both reservoirs unioned into one polygon record is shown here.



But isn’t this a really UGLY outline??

